Import the data:

In[4]:=
```
rxpkp = Import["~/Desktop/underfive.csv"][[2 ;;]];
```

In[5]:=
```
rxpkp[[Range[1, 10]]]
```

Out[5]=
```
{{0, 177.1}, {1, 180.679}, {2, 181.387}, {3, 162.501}, {4, 138.025},
 {5, 107.407}, {6, 84.2726}, {7, 76.7897}, {8, 80.8254}, {9, 109.721}}
```

Plot the data:

In[6]:=
```
ListPlot[rxpkp, Frame → {True, True, False, False},
 FrameLabel → {"Months since 1 Jan 2011",
   "Prescriptions per 1,000 beneficiaries"}, BaseStyle → FontSize → 12]
```

Out[6]=



Fit the sinusoid:

$\beta 0$: intercept of the upper trend line (peak monthly prescribing rate per 1,000 beneficiaries)

$\beta 1$: slope of the upper trend line (peak monthly prescribing rate per 1,000 beneficiaries)

$\gamma 0$: intercept of the lower trend line (trough monthly prescribing rate per 1,000 beneficiaries)

$\gamma 1$: slope of the lower trend line (trough monthly prescribing rate per 1,000 beneficiaries)

δ: phase shift of the sinusoid

In[7]:=
```
upper[x_] := β0 + β1 x ; (*The upper trend line*)
lower[x_] := γ0 + γ1 x ; (*The lower trend line*)
```

In[9]:=
```
fit = NonlinearModelFit[rxpkp, {
    1/2 (upper[x] + lower[x]) + 1/2 (upper[x] - lower[x]) (Cos[2 π/12 (x - δ)])
   }, {β0, β1, γ0, γ1, δ}, x];
```

In[10]:=
```
fit
```

Out[10]=
```
FittedModel[ 1/2 (274.494 - 2.21045 x) + 1/2 (108.275 - 1.16668 x) Cos[1/6 π (-0.658059 + x)] ]
```

In[11]:=
```
Normal[fit]
```

Out[11]=
$$\frac{1}{2}\left(274.494 - 2.21045\,x\right) + \frac{1}{2}\left(108.275 - 1.16668\,x\right)\text{Cos}\left[\frac{1}{6}\pi\left(-0.658059 + x\right)\right]$$

In[12]:=
```
fit["BestFitParameters"]
```

Out[12]=
$$\{\beta 0 \to 191.384,\ \beta 1 \to -1.68856,\ \gamma 0 \to 83.1096,\ \gamma 1 \to -0.521889,\ \delta \to 0.658059\}$$

In[13]:=
```
fit["ParameterConfidenceIntervalTable"]
```

Out[13]=

	Estimate	Standard Error	Confidence Interval
$\beta 0$	191.384	5.61896	{180.109, 202.659}
$\beta 1$	−1.68856	0.181924	{−2.05362, −1.32351}
$\gamma 0$	83.1096	6.06145	{70.9464, 95.2727}
$\gamma 1$	−0.521889	0.176705	{−0.876474, −0.167303}
δ	0.658059	0.123528	{0.410181, 0.905937}

Plot the output:

In[14]:=
```
fitplot = Plot[fit[x], {x, rxpkp[[1, 1]], rxpkp[[-1, 1]]},
   PlotRange → All, PlotStyle → Black, PlotLegends → {"Sinusoid fit"}];
```

In[15]:=
```
upperplot = Plot[upper[x] /. fit["BestFitParameters"],
   {x, rxpkp[[1, 1]], rxpkp[[-1, 1]]}, PlotStyle → Blue, PlotLegends → {"Envelope"}];
lowerplot = Plot[lower[x] /. fit["BestFitParameters"],
   {x, rxpkp[[1, 1]], rxpkp[[-1, 1]]}, PlotStyle → Blue];
```

In[17]:=
```
dataplot = ListLinePlot[rxpkp, PlotMarkers → Automatic,
   PlotStyle → {Gray}, PlotRange → All, PlotLegends → {"Data"}];
```

In[18]:=
```
figunder5sinusoid = Show[dataplot, fitplot, upperplot, lowerplot,
  PlotRange → {0, Automatic}, Frame → {True, True, False, False},
  FrameLabel → {None, "Prescriptions per 1,000\nbeneficiaries"},
  BaseStyle → FontSize → 14, ImageSize → 400,
  FrameTicks -> {{{0, "2011"}, {12, "2012"}, {24, "2013"},
     {36, "2014"}, {48, "2015"}}, Automatic},
  GridLines → {Table[i, {i, 0, 54, 6}], {50, 100, 150, 200}}]
```

Out[18]=